

Comment gello

GELLO can have the context of an IncomingObservation, for example a pathology result. Metadata within the GELLO can be set to specify LOINC codes for a host system to watch for and if present, run the GELLO.

This worked example will demonstrate using screen shots from the Medical Objects Explorer software. It will show some GELLO that looks for archetyped clinical patient data of height , weight, waist circumference and Blood pressure, such as might be obtained when a patient presents to a medical clinic. The code will take this data, and also look for recent blood tests results; and run its logic as to whether certain conditions are present such as being overweight or having abnormal metabolic profiles. We will use BMI and Triglyceride/HDL ratio. The result of the GELLO is another observation that presents in the UI as System Generated information, below the displayed Observation that has triggered the Comment GELLO.

The context is:

```
Context HL7_v2_VMR_V1::IncomingObservation
```

Metadata is a tuple we create as a local variable and is customisable. MO Explorer knows to look for this when the GELLO is uploaded to the server it uses and stores the Comment GELLO for clinical use.

Here's some metadata:

```
Let metadata = Tuple {
    title = 'calculate metabolic comments on admission vitals and recent pathology',
    keywords = Sequence{'BMI', 'Weight', 'Metabolic'},
    version = 1,
    revision = 16,
    date = '20180227',
    author = 'Medical-Objects',
    identifier = 'COMMENT_GELLO:CALCULATE_Metabolic_Comments',
    purpose = 'For metabolic calculation after basic admission and registration vitals
taken by nursing staff',
    includesList = Sequence{'29463-7^^LN', '8302-2^^LN', '8280-0^^LN', '8480-6^^LN', '8462-
4^^LN'},
    excludesList = Sequence{}}
```

Now we can set up some codes for the retrievable data:

```
Let measured_weight_LN: CD = CD{code = '29463-7', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
Let measured_height_LN: CD = CD{code = '8302-2', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
Let measured_waist_LN: CD = CD{code = '8280-0', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
Let sys_BP_LN: CD = CD{code = '8480-6', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
Let diast_BP_LN: CD = CD{code = '8462-4', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
```

These will be the codes that the archetyped data is mapped to.

Add the required lab data LOINC codes:

```
Let trig_LN: CD = CD{code = '14927-8', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
Let hdl_LN: CD = CD{code = '14646-4', codeSystem = '2.16.840.1.113883.6.1', codeSystemName ='LN'}
```

Now we get the observation data and the vmr data:

```
-- get the observations and vmr data
Let currentWeight:PQ = observations->select(o| o.observationCode.implies(measured_weight_LN).value)->sortedBy
(dateTime)->last().value.oclAsType(PQ)
Let currentHeight:PQ = observations->select(o| o.observationCode.implies(measured_height_LN).value)->sortedBy
(dateTime)->last().value.oclAsType(PQ)
Let currentWaist:PQ = observations->select(o| o.observationCode.implies(measured_waist_LN).value)->sortedBy
(dateTime)->last().value.oclAsType(PQ)
Let last_trigObservation: Observation = If vmr.observations->select(o| o.observationCode.implies(trig_LN).
value)->sortedBy(dateTime)->last().oclIsDefined() then
    vmr.observations->select(o| o.observationCode.implies(trig_LN).value)->sortedBy(dateTime)->last() else
null endif
Let last_hdlObservation: Observation = If vmr.observations->select(o| o.observationCode.implies(hdl_LN).
value)->sortedBy(dateTime)->last().oclIsDefined() then
    vmr.observations->select(o| o.observationCode.implies(hdl_LN).value)->sortedBy(dateTime)->last() else null
endif
Let last_sysBPObservation: Observation = If observations->select(o| o.observationCode.implies(sys_BP_LN).
value)->sortedBy(dateTime)->last().oclIsDefined() then
    observations->select(o| o.observationCode.implies(sys_BP_LN).value)->sortedBy(dateTime)->last() else null
endif
Let last_diastBPObservation: Observation = If observations->select(o| o.observationCode.implies(diast_BP_LN).
value)->sortedBy(dateTime)->last().oclIsDefined() then
    observations->select(o| o.observationCode.implies(diast_BP_LN).value)->sortedBy(dateTime)->last() else
null endif
```

Notice that for the data obtained from the IncomingObservation ie the measured height, weight, waist circumference, systolic blood pressure and diastolic blood pressure; we don't need to say 'vmr.observations', but when we are looking outside the context of the incoming observation, we do- e.g. for recent pathology. This code keeps the convention of searching for Observations, then ordering them by date/Time, then getting the last Observation. This is probably unnecessary for the archetyped clinical data, but it is possible to repeat a blood pressure reading if high ; so it is retained.

Do a BMI calculation: (we know the units will be cm for height and kg for weight.

```
-- do BMI calculation
Let currentHeightInMeters: PQ = currentHeight/100
Let bmi: Real = currentWeight.value/currentHeightInMeters.value.power(2)
Let bmi_Rounded: Real = bmi.round()
Let bmi_AsString: String = If bmi_Rounded.octIsDefined() then bmi_Rounded.toChar() else '' endif
```

Do a triglyceride / HDL ratio:

```
--TG HDL ratio
Let trig_HDL_Ratio : Real = If last_trigObservation.octIsDefined() and last_hdlObservation.octIsDefined()
then
    last_trigObservation.value.octAsType(PQ).value / last_hdlObservation.value.octAsType(PQ).value
else null endif
Let trig_HDL_Ratio_raised: Boolean = If last_trigObservation.octIsDefined() and last_hdlObservation.
octIsDefined() then
    If last_trigObservation.value.octAsType(PQ).value / last_hdlObservation.value.octAsType(PQ).value > 2
    then True
    else False endif
else False endif
```

Prepare comments:

```
-- prepare comments
Let commentCode:CD = CD{code = '8251-1', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN',
displayName = ST{value='Comment'}}
Let bmiString: String = 'BMI is '
Let additionalBMIComment: String = If bmi.octIsDefined() then
    If bmi <18.6 then '. Patient is considered to be Underweight.' else
    If bmi >25 and bmi < 30 then '. Patient is considered to be Overweight.' else
    If bmi >29.9 then '. Patient is considered to be Obese.' else
    ''
endif endif endif else '' endif
Let trig_HDL_ratio_asChar: String = If trig_HDL_Ratio.octIsDefined() then trig_HDL_Ratio.toChar() else ''
endif

Let trig_HDL_ratio_comment1: String = If trig_HDL_Ratio_raised then
    '\nTriglyceride-HDL ratio is ' else '' endif
Let trig_HDL_ratio_comment2: String = If trig_HDL_Ratio_raised then trig_HDL_ratio_asChar.concat('. Levels
above 2 are significant.')
else '' endif
Let trig_HDL_ratio_comment: String = trig_HDL_ratio_comment1.concat(trig_HDL_ratio_comment2)

Let comment: String = bmiString.concat(bmi_AsString).concat(additionalBMIComment).concat
(trig_HDL_ratio_comment)
```

Produce the result:

```
-- result
Let result_comment:ST = ST{value=comment}
Let result:Sequence(Observation)
    = Sequence{Observation{
        value = result_comment,
        observationCode = commentCode
    }
}
result
```

OK! So here's the input screen in MO Explorer, for the archetype that becomes the Incoming Observation that the GELLO is watching for the LOINC codes for (Comment GELLO just as easily runs against HL7 lab data Observations):

Mrs Samantha PATIENT		Born 13-Aug-1981 (36y) Gender Female		
Address		Phone	Not recorded	Medicare No
Specimen	Lab No BC425D7E-89B2-4CD2-81FE-B05F21DD987D	Request Date 5/02/2018	Effective Date 5/02/2018	Generated Date 5/02/2018 11:27 AM
Requested By	PETER SCOTT	CC		

Creating CEN.Basic_Admission_Registration_Obs.v1 Lab No: "" for Mrs Samantha Patient

Height	<input type="text"/>	Units	cm
Weight	<input type="text"/>	Units	kg
Waist circumference	<input type="text"/>	Units	cm
Blood pressure			
Systolic BP	<input type="text"/>	Units	mmHg
Diastolic BP	<input type="text"/>	Units	mmHg

OK Cancel

Here's what happens when we run the gello explicitly against this patient's data:

Medical-Objects GELLO IDE (Mowgli)

File Edit Search View Run Build Tools Help

Test

```

1 Context HL7_v2_VMR_V1::IncomingObservation
  + 2 let metadata = Tuple {
      3   title = 'calculate metabolic comments on admission vitals and recent pathology',
      4   keywords = Sequence('BMI', 'Weight', 'Metabolic'),
      5   version = 1,
      6   revision = 1,
      7   date = '20180227',
      8   author = 'Medical-Objects',
      9   identifier = 'COMMENT_GELLO:Test',
     10  purpose = 'For metabolic calculation after basic admission and registration vitals taken by'
     11  includesList = Sequence('29463-7~LN', '8302-2~LN', '8280-0~LN', '8460-6~LN', '8462-4~LN')
     12  excludesList = Sequence()
    }

    Let measured_weight_LN: CD = CD(code = '29463-7', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let measured_height_LN: CD = CD(code = '8302-2', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let measured_weight_CD: CD = CD(code = '8280-0', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let syst_BP_LN: CD = CD(code = '8462-4', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let diast_BP_LN: CD = CD(code = '14927-8', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let trig_LN: CD = CD(code = '14646-8', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')
    Let hdL_IN: CD = CD(code = '14646-8', codeSystem = '2.16.840.1.113883.6.1', codeSystemName = 'LN')

    -- get the observations and vmr data
    Let currentHeight: PQ = observations->select(o) o.observationCode.implies(measured_weight_LN).value->sortedBy(date)
    Let currentWeight: PQ = observations->select(o) o.observationCode.implies(measured_height_LN).value->sortedBy(date)
    Let currentWaist: PQ = observations->select(o) o.observationCode.implies(measured_weight_LN).value->sortedBy(date)
    Let last_trigObservation: Observation = If vmr.observations->select(o) o.observationCode.implies(trig_LN).value->last() else null endi
    Let last_hdLObservation: Observation = If vmr.observations->select(o) o.observationCode.implies(hdL_IN).value->last() else null endi
    Let last_diastPObservation: Observation = If observations->select(o) o.observationCode.implies(diast_BP_LN).value->last() else null endi
    Let last_systPObservation: Observation = If observations->select(o) o.observationCode.implies(syst_BP_LN).value->last() else null endi
    Let last_hdLPObservation: Observation = If observations->select(o) o.observationCode.implies(hdL_IN).value->last() else null endi
    Let last_diastPObservation: Observation = If observations->select(o) o.observationCode.implies(diast_BP_LN).value->last() else null endi
    Let last_systPObservation: Observation = If observations->select(o) o.observationCode.implies(syst_BP_LN).value->last() else null endi

    -- do BMI calculation
    Let currentHeightInMeters: PQ = currentHeight/100
    Let bmi: Real = currentWeight.value/currentHeightInMeters.value.power(2)
    Let bmi_Rounded: Real = bmi.round()
    Let bmi_AsString: String = If bmi_Rounded.oclisDefined() then bmi_Rounded.toChar() else '' endif
    --TO HDL ratio
    Let trig_HDL_Ratio : Real = If last_trigObservation.oclisDefined() and last_hdLObservation.oclisDefined() then
      last_trigObservation.value.oclisType(PQ).value / last_hdLObservation.value.oclisType(PQ).value
    else null endif
    Let trig_HDL_Ratio_raised: Boolean = If last_trigObservation.oclisDefined() and last_hdLObservation.oclisDefined() then
      If last_trigObservation.value.oclisType(PQ).value / last_hdLObservation.value.oclisType(PQ).value > 2
      then True
      else False endif
  
```

Results Explorer

Name	Class	Data
currentHeight	PQ	167 cm
currentWeight	PQ	100 kg
last_trigObservation	Observation	<Observation: TObservation>
last_hdLObservation	Observation	<Observation: TObservation>
last_systPObservation	Observation	<Observation: TObservation>
last_diastPObservation	Observation	<Observation: TObservation>
currentHeightInMeters	PQ	1.67 m
bmi	Real	31.912234572771
bmi_Rounded	Integer	32
bmi_AsString	String	32
trig_HDL_ratio_raised	Real	5
commentCode	CD	Comment
bmString	String	BMI is
additionalBMComment	String	. Patient is considered to be Obese.
trig_HDL_ratio_eChar	String	5
trig_HDL_ratio_comment1	String	Triglyceride+HDL ratio is
trig_HDL_ratio_comment2	String	5. Levels above 2 are significant.
trig_HDL_ratio_comment	String	Triglyceride+HDL ratio is 5. Levels above 2 are significant.
comment	String	BMI is 32. Patient is considered to be Obese.
result_comment	ST	Triglyceride+HDL ratio is 5. Levels above 2 are significant.
result[1]	Sequence(Observation)	BMI is 32. Patient is considered to be Obese. Triglyceride+HD...
Result[1]	Sequence(Observation)	Sequence<Observation: TObservation>
[1]	Observation	Sequence<Observation: TObservation>
id	String	<Observation: TObservation>
encounter	Encounter	<null>
observationCode	CD	Comment
components	Sequence(Observation)	<null>
identifier	II	<null>
value	ST	BMI is 32. Patient is considered to be Obese. Triglyceride+HD...
referenceRange	IVL_QTY	<null>
abnormalFlag	CD	<null>
dateTime	TS	<null>
diagnosticServiceSectionID	CD	<null>
specimenCode	CD	<null>
templateID	String	<null>

Model Explorer | Data Explorer (IncomingObservation) | Results Explorer

Stack

Name	Class	Data

Here's the UI of the Observation showing the comment as a string under 'System Generated information':

File Patients Documents Create Document Providers View Navigate Tools Reports Decision Support Window Help Dev Tools

Mrs Samantha PATIENT Born: 13-Aug-1981 (36y) Gender: Female

Address	Specimen Lab No: 16045313-97BA-401F-A04D-3165142CFCDC	Phone: Not recorded	Medicare No:
		Request Date: 20/02/2018	Effective Date: 20/02/2018
			Generated Date: 20/02/2018 9:30 AM
		CC:	
Requested By: DR PETER SCOTT			

Cumulative Results Park Result Request Chart Path Request Create Response Provider Note Result History Edit Report

CEN.Basic_Admission_Registration_Obs.v1 Basic view active:

	Units	Time
Height	cm	20.2.2018
Weight	kg	20.2.2018
Waist circumference	cm	20.2.2018
BLOOD PRESSURE	20.2.2018	
Systolic BP	mmHg	20.2.2018
Diastolic BP	mmHg	20.2.2018

System Generated Information
BMI is 32. Patient is considered to be Obese.
Triglyceride-HDL ratio is 5. Levels above 2 are significant.

Report Author: Dr. PETER SCOTT Service Provider: Peter Scott Demo Equator
From Peter Scott Demo Equator "MOE02200930544-4839" 20.02.2018

Available Results
Diseases Groups Graphs
0.02.2018 : CEN-ELFTS.v1
0.02.2018 : CEN_Basic_Admission_Registra
5.02.2018 : CEN_Basic_Admission_Registra
1.10.2017 : Diet followup
5.10.2017 : Lipid Measurement
4.10.2017 : Lipid Measurement
4.10.2017 : Diet followup
3.10.2017 : Diet followup
5.10.2016 : Iron measurement
5.09.2016 : Update Health Record
3.07.2016 : Full Blood Count
5.05.2016 : Electrolytes and LFTs
5.05.2016 : osmolality - urine
5.05.2016 : Thyroid panel
5.05.2016 : Electrolytes and LFTs
4.05.2016 : Patient Registration
4.05.2016 : Patient Registration